

# The Trie Data Structure<sup>\*</sup>

## Topics

- Purely functional programming
- Algebraic data types
- Polymorphism

## Introduction

In this series of exercises, we will learn to build tries. A trie is a form of persistent maps that associates list like keys to values.

For starters, we will build a degenerate version of the structure, immutable sets of words. Then we will associate values to words, and finally generalize to list keys instead of words.

**Note:** for the sake of simplicity, we will only use lists and polymorphic comparison for implementing this data structure, but the resulting complexity is not the best one could expect for tries.

## Exercise 1 – dictionnaires

We will build sets of words (dictionnaires), with two primitives: present and insert (feel free to add others).

The (recursive) type for dictionnaires is:

```
1 type dict = Node of bool * (char * dict) list
```

It is a  $n$ -ary tree structure in which the Node constructor represents a node in the tree and the list the tagged edges to its children. Each node bears a bool, that indicates if the word whose letters are the tags of the edges from the root is a member or the set.

**Bonus:** here is a little program to test your code.

```
1 let rec main dict =
2   match Str.(split (regexp "[_\\t]+") (read_line ())) with
3   | [ "present" ; word ] ->
4     print_endline (if present word dict then "Present." else "Not_present.");
5     main dict
6   | "insert" :: words ->
7     let dict = List.fold_right insert words dict in
8     print_endline "Added." ;
9     main dict
10  | "exit" ->
11    print_endline "Bye."
12  | _ ->
13    print_endline "Unknown_command." ;
14    main dict
15 let () = main empty
```

**Question 1.1** – Write an auxiliary function `explode` that turns a string into a list of characters. Use an internal tail recursive function (Bonus if you don't `List.rev`).

**Question 1.2** – Write a value `empty` of type `dict`, representing the empty dictionary (of which even the empty string is not a member).

**Question 1.3** – Write `present` of type `string -> dict -> bool`, that tells if a word is present in the dictionary.

**Question 1.4** – Write `present` of type `string -> dict -> dict`, that returns a new dictionary with an added word.

## Exercise 2

We now generalize the dictionary to a type `'a dictmap` that associates values of type `'a` to words.

**Question 2.1** – Define the type `'a dictmap`.

**Question 2.2** – Write `empty` of type `'a dictmap`.

**Question 2.3** – Write `get` of type `string -> 'a dictmap -> 'a` that returns the values associated to a word or fails with `Not_found`.

**Question 2.4** – Write `set` of type `string -> 'a -> 'a dictmap -> 'a dictmap` that initializes or updates the value associated to a word.

**Question 2.5** – Update the test program to test your implementation of `string dictmap`.

## Exercise 3

**Question 3.1** – Same questions, but generalizes to a type `('a, 'b) trie` that associates values of type `'b` to keys of type `'a list`.

**Question 3.2** – Update the test program to test your implementation of `(char, string) dictmap`.

# The Trie Data Structure

SOLUTIONS – SOLUTIONS – SOLUTIONS – SOLUTIONS

## Solution to question 1.1

```
1 let explode s =  
2   let rec loop acc i =  
3     if i < 0 then acc else loop (String.get s i :: acc) (pred i) in  
4   loop [] (String.length s - 1)
```

## Solution to question 1.2

```
1 let empty : dict = Node (false, [])
```

## Solution to question 1.3

```
1 let present : string -> dict -> bool = fun s dict ->  
2   let rec descend = function  
3     | [], Node (present, _) -> present  
4     | _ :: _, Node (_, []) -> false  
5     | c :: cs as s, Node (present, (subc, subdict) :: rest) ->  
6       if c = subc then  
7         descend (cs, subdict)  
8       else  
9         descend (s, Node (present, rest)) in  
10  descend (explode s, dict)
```

## Solution to question 1.4

```
1 let insert : string -> dict -> dict = fun s dict ->  
2   let rec descend = function  
3     | [], Node (_, subs) -> Node (true, subs)  
4     | c :: cs, Node (present, []) ->  
5       Node (present, [ (c, descend (cs, empty)) ])  
6     | c :: cs as s, Node (present, (subc, subdict) :: rest) ->  
7       if c = subc then  
8         Node (present, (subc, descend (cs, subdict)) :: rest)  
9       else  
10        let Node (_, nrest) = descend (s, Node (present, rest)) in  
11        Node (present, (subc, subdict) :: nrest)  
12  in  
13  descend (explode s, dict)
```

## Solution to question 2.1

```
1 type 'a dictmap = Node of 'a option * (char * 'a dictmap) list
```

## Solution to question 2.2

```
1 let empty : 'a dictmap = Node (None, [])
```

### Solution to question 2.3

```
1 let get : string -> 'a dictmap -> 'a = fun s dict ->
2   let rec descend = function
3     | [], Node (Some value, _) -> value
4     | [], Node (None, _) -> raise Not_found
5     | _ :: _, Node (_, []) -> raise Not_found
6     | c :: cs as s, Node (present, (subc, subdict) :: rest) ->
7       if c = subc then
8         descend (cs, subdict)
9       else
10        descend (s, Node (present, rest)) in
11  descend (explode s, dict)
```

### Solution to question 2.4

```
1 let set : string -> 'a -> 'a dictmap -> 'a dictmap = fun s v dict ->
2   let rec descend = function
3     | [], Node (_, subs) -> Node (Some v, subs)
4     | c :: cs, Node (present, []) ->
5       Node (present, [ (c, descend (cs, empty)) ])
6     | c :: cs as s, Node (present, (subc, subdict) :: rest) ->
7       if c = subc then
8         Node (present, (subc, descend (cs, subdict)) :: rest)
9       else
10        let Node (_, nrest) = descend (s, Node (present, rest)) in
11        Node (present, (subc, subdict) :: nrest)
12  in
13  descend (explode s, dict)
```

### Solution to question 2.5

```
1 let rec main dict =
2   match Str.split (regexp "[_\\t]+") (read_line ()) with
3   | [ "get" ; word ] ->
4     begin try
5       print_endline (get word dict)
6     with Not_found ->
7       print_endline "Not_found."
8     end ;
9     main dict
10  | [ "set" ; word ; value ] ->
11    let dict = set word value dict in
12    print_endline "Added." ;
13    main dict
14  | [ "exit" | "quit" | "bye" ] ->
15    print_endline "Bye."
16  | _ ->
17    print_endline "Unknown_command." ;
18    main dict
19
20 let () = main empty
```

### Solution to exercise 3

Nothing much to change to the code actually, except for lifting the calls to explode and updating the type definition and annotations.

```

1 type ('a, 'b) trie = Node of 'b option * ('a * ('a, 'b) trie) list
2
3 let empty = Node (None, [])
4
5 let get = fun s dict ->
6   let rec descend = function
7     | [], Node (Some value, _) -> value
8     | [], Node (None, _) -> raise Not_found
9     | _ :: _, Node (_, []) -> raise Not_found
10    | c :: cs as s, Node (present, (subc, subdict) :: rest) ->
11      if c = subc then
12        descend (cs, subdict)
13      else
14        descend (s, Node (present, rest)) in
15    descend (s, dict)
16
17 let set = fun s v dict ->
18   let rec descend = function
19     | [], Node (_, subs) -> Node (Some v, subs)
20     | c :: cs, Node (present, []) ->
21       Node (present, [ (c, descend (cs, empty)) ])
22     | c :: cs as s, Node (present, (subc, subdict) :: rest) ->
23       if c = subc then
24         Node (present, (subc, descend (cs, subdict)) :: rest)
25       else
26         let Node (_, nrest) = descend (s, Node (present, rest)) in
27         Node (present, (subc, subdict) :: nrest)
28   in
29     descend (s, dict)
30
31 let explode s =
32   let rec loop acc i =
33     if i < 0 then acc else loop (String.get s i :: acc) (pred i) in
34   loop [] (String.length s - 1)
35
36 let rec main dict =
37   match Str.(split (regexp "[_\\t]+") (read_line ())) with
38   | [ "get" ; word ] ->
39     begin try
40       print_endline (get (explode word) dict)
41     with Not_found ->
42       print_endline "Not_found."
43     end ;
44     main dict
45   | [ "set" ; word ; value ] ->
46     let dict = set (explode word) value dict in
47     print_endline "Added." ;
48     main dict
49   | [ "exit" | "quit" | "bye" ] ->
50     print_endline "Bye."
51   | _ ->
52     print_endline "Unknown_command." ;
53     main dict

```

54

55 **let** () = main empty